Relaxation modulus of a polyurethane:




| $\log(t, \mathrm{min})$ | $E_{rel}(t)$, psi |
|---|---|
| -6 | 56,280 |
| -5 | 22,880 |
| -4 | 4,450 |
| -3 | 957 |
| -2 | 578 |
| -1 | 481 |
| 0 | 480 |

# The Wiechert Model




$$\bar{\sigma} = \bar{\sigma}_e + \sum_j \bar{\sigma}_j$$

$$= \left\{ k_e + \sum_j \frac{k_j s}{s + \frac{1}{\tau_j}} \right\} \bar{\epsilon}$$

$$\bar{\sigma} = \mathcal{E}\,\bar{\epsilon}$$

Relaxation: $\bar{\epsilon} = \frac{\epsilon_0}{s} \rightarrow \frac{\bar{\sigma}}{\epsilon_0} = \bar{E}_{rel} = \frac{\mathcal{E}}{s}$

$$\bar{E}_{rel} = \frac{k_e}{s} + \sum_j \frac{k_j}{s + \frac{1}{\tau_j}}$$

$$E_{rel}(t) = k_e + \sum_j k_j e^{-t/\tau_j}$$

# Shapery Collocation

Glassy and rubbery moduli:

```
> E_g:=91100;E_r:=480;
```

$$E\_g := 91100$$
$$E\_r := 480$$

Arrays of time and relaxation time:

```
> t:=array(1..6,[10^(-6),10^(-5),10^(-4),10^(-3),10^(-2),10^(-1)]);
```

$$t := \left[\frac{1}{1000000}, \frac{1}{100000}, \frac{1}{10000}, \frac{1}{1000}, \frac{1}{100}, \frac{1}{10}\right]$$

```
> tau:=array(1..6,[10^(-6),10^(-5),10^(-4),10^(-3),10^(-2),10^(-1)])
  ;
```

$$\tau := \left[\frac{1}{1000000}, \frac{1}{100000}, \frac{1}{10000}, \frac{1}{1000}, \frac{1}{100}, \frac{1}{10}\right]$$

Coefficient matrix **A** in **Ak=B**

```
> ke:=E_r;A:=array(1..6,1..6);
```

$$ke := 480$$
$$A := \mathrm{array}(1\,..\,6, 1\,..\,6, [\;])$$

```
> for i from 1 to 6 do
> for j from 1 to 6 do
>   A[i,j]:=exp(-t[i]/tau[j]);if (evalf(A[i,j])<.01) then A[i,j]:=0
  fi
> od;od;
> Digits:=4:'A'=evalf(map(eval,A));
```

$$A = \begin{bmatrix} .3679 & .9048 & .9900 & .9990 & .9999 & 1.000 \\ 0 & .3679 & .9048 & .9900 & .9990 & .9999 \\ 0 & 0 & .3679 & .9048 & .9900 & .9990 \\ 0 & 0 & 0 & .3679 & .9048 & .9900 \\ 0 & 0 & 0 & 0 & .3679 & .9048 \\ 0 & 0 & 0 & 0 & 0 & .3679 \end{bmatrix}$$

```
> with(linalg):
```

Inverse of coefficient matrix

```
> A_inv:=evalf(map(eval,inverse(A)));
```

$$A\_inv := \begin{bmatrix} 2.718 & -6.682 & 9.127 & -11.83 & 15.33 & -19.97 \\ 0 & 2.718 & -6.682 & 9.127 & -11.83 & 15.33 \\ 0 & 0 & 2.718 & -6.682 & 9.127 & -11.83 \\ 0 & 0 & 0 & 2.718 & -6.682 & 9.127 \\ 0 & 0 & 0 & 0 & 2.718 & -6.682 \\ 0 & 0 & 0 & 0 & 0 & 2.718 \end{bmatrix}$$

rhs vector **B**:

```
> Er:=vector(6,[56280,22880,4450,957,578,481]);
```

$$Er := [56280, 22880, 4450, 957, 578, 481]$$

```
> B:=evalm(Er-ke);
```

$$B := [55800, 22400, 3970, 477, 98, 1]$$

multiply **A**inverse by **B** to get **k** values

```
> k:=array(1..6);
```

$$k := \text{array}(1..6, [\ ])$$

```
> k:=evalm(A_inv &* B);
```

$$k := [34070., 37560., 8485., 650.3, 259.7, 2.718]$$

Correct for model undershoot:

```
> undershoot:=E_g-(ke+sum('k[i]','i'=1..6));
```

$$undershoot := 9590.$$

```
> k[1]:=k[1]+undershoot;
```

$$k_1 := 43660.$$

```
> 'k_final'=evalm(k);
```

$$k\_final = [43660., 37560., 8485., 650.3, 259.7, 2.718]$$

Examine and plot final model formulation:

```
> E_rel:=ke+sum('k[j]'*exp(-10^log_t/'tau[j]'),'j'=1..6);
```

$$E\_rel := 480 + 43660.\ \mathbf{e}^{(-1000000\ 10^{log\_t})} + 37560.\ \mathbf{e}^{(-100000\ 10^{log\_t})} + 8485.\ \mathbf{e}^{(-10000\ 10^{log\_t})}$$
$$+ 650.3\ \mathbf{e}^{(-1000\ 10^{log\_t})} + 259.7\ \mathbf{e}^{(-100\ 10^{log\_t})} + 2.718\ \mathbf{e}^{(-10\ 10^{log\_t})}$$

```
> plot(log10(E_rel),log_t=-8..0);
```





$E_{rel}(t) = k_e + \sum_j k_j e^{-t/\tau_j}$
$\rightarrow k_e + \int k(\tau) e^{-t/\tau} d\tau$
5
4
3
2
1
0
-6
-5
-4
-3
-2
-1
log [k(t), psi]
log (τ, s)